

# NAG Fortran Library Routine Document

## F04CCF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

F04CCF computes the solution to a complex system of linear equations  $AX = B$ , where  $A$  is an  $n$  by  $n$  tridiagonal matrix and  $X$  and  $B$  are  $n$  by  $r$  matrices. An estimate of the condition number of  $A$  and an error bound for the computed solution are also returned.

### 2 Specification

```

SUBROUTINE F04CCF (N, NRHS, DL, D, DU, DU2, IPIV, B, LDB, RCOND, ERBND,
1                IFAIL)
INTEGER          N, NRHS, IPIV(*), LDB, IFAIL
double precision RCOND, ERBND
complex*16      DL(*), D(*), DU(*), DU2(*), B(LDB,*)

```

### 3 Description

The  $LU$  decomposition with partial pivoting and row interchanges is used to factor  $A$  as  $A = PLU$ , where  $P$  is a permutation matrix,  $L$  is unit lower triangular with at most one non-zero sub-diagonal element, and  $U$  is an upper triangular band matrix with two super-diagonals. The factored form of  $A$  is then used to solve the system of equations  $AX = B$ .

Note that the equations  $A^T X = B$  may be solved by interchanging the order of the arguments  $DU$  and  $DL$ .

### 4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: <http://www.netlib.org/lapack/lug>

Higham N J (2002) *Accuracy and Stability of Numerical Algorithms* (2nd Edition) SIAM, Philadelphia

### 5 Parameters

- 1: N – INTEGER *Input*  
*On entry:* the number of linear equations  $n$ , i.e., the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 2: NRHS – INTEGER *Input*  
*On entry:* the number of right-hand sides  $r$ , i.e., the number of columns of the matrix  $B$ .  
*Constraint:*  $NRHS \geq 0$ .
- 3: DL(\*) – **complex\*16** array *Input/Output*  
**Note:** the dimension of the array  $DL$  must be at least  $\max(1, N - 1)$ .  
*On entry:*  $DL$  must contain the  $(n - 1)$  sub-diagonal elements of the matrix  $A$ .  
*On exit:* if  $IFAIL \geq 0$ ,  $DL$  is overwritten by the  $(n - 1)$  multipliers that define the matrix  $L$  from the  $LU$  factorization of  $A$ .

- 4:  $D(*)$  – **complex\*16** array *Input/Output*  
**Note:** the dimension of the array  $D$  must be at least  $\max(1, N)$ .  
*On entry:*  $D$  must contain the  $n$  diagonal elements of the matrix  $A$ .  
*On exit:* if  $IFAIL \geq 0$ ,  $D$  is overwritten by the  $n$  diagonal elements of the upper triangular matrix  $U$  from the  $LU$  factorization of  $A$ .
- 5:  $DU(*)$  – **complex\*16** array *Input/Output*  
**Note:** the dimension of the array  $DU$  must be at least  $\max(1, N - 1)$ .  
*On entry:*  $DU$  must contain the  $(n - 1)$  super-diagonal elements of the matrix  $A$   
*On exit:* if  $IFAIL \geq 0$ ,  $DU$  is overwritten by the  $(n - 1)$  elements of the first super-diagonal of  $U$ .
- 6:  $DU2(*)$  – **complex\*16** array *Output*  
**Note:** the dimension of the array  $DU2$  must be at least  $\max(1, N - 2)$ .  
*On exit:* if  $IFAIL \geq 0$ ,  $DU2$  returns the  $(n - 2)$  elements of the second super-diagonal of  $U$ .
- 7:  $IPIV(*)$  – INTEGER array *Output*  
**Note:** the dimension of the array  $IPIV$  must be at least  $\max(1, N)$ .  
*On exit:* if  $IFAIL \geq 0$ , the pivot indices that define the permutation matrix  $P$ ; at the  $i$ th step row  $i$  of the matrix was interchanged with row  $IPIV(i)$ .  $IPIV(i)$  will always be either  $i$  or  $(i + 1)$ ;  $IPIV(i) = i$  indicates a row interchange was not required.
- 8:  $B(LDB,*)$  – **complex\*16** array *Input/Output*  
**Note:** the second dimension of the array  $B$  must be at least  $\max(1, NRHS)$ . To solve the equations  $Ax = b$ , where  $b$  is a single right-hand side,  $B$  may be supplied as a one-dimensional array with length  $LDB = \max(1, N)$ .  
*On entry:* the  $n$  by  $r$  matrix of right-hand sides  $B$ .  
*On exit:* if  $IFAIL = 0$  or  $N + 1$ , the  $n$  by  $r$  solution matrix  $X$ .
- 9:  $LDB$  – INTEGER *Input*  
*On entry:* the first dimension of the array  $B$  as declared in the (sub)program from which F04CCF is called.  
*Constraint:*  $LDB \geq \max(1, N)$ .
- 10:  $RCOND$  – **double precision** *Output*  
*On exit:* if  $IFAIL \geq 0$ , an estimate of the reciprocal of the condition number of the matrix  $A$ , computed as  $RCOND = 1 / (\|A\|_1 \|A^{-1}\|_1)$ .
- 11:  $ERRBND$  – **double precision** *Output*  
*On exit:* if  $IFAIL = 0$  or  $N + 1$ , an estimate of the forward error bound for a computed solution  $\hat{x}$ , such that  $\|\hat{x} - x\|_1 / \|x\|_1 \leq ERRBND$ , where  $\hat{x}$  is a column of the computed solution returned in the array  $B$  and  $x$  is the corresponding column of the exact solution  $X$ . If  $RCOND$  is less than **machine precision**, then  $ERRBND$  is returned as unity.
- 12:  $IFAIL$  – INTEGER *Input/Output*  
*On entry:*  $IFAIL$  must be set to 0,  $-1$  or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.  
*On exit:*  $IFAIL = 0$  unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or  $1$  is recommended. If the output of error messages is undesirable, then the value  $1$  is recommended. Otherwise, for users not familiar with this parameter the recommended value is  $0$ . **When the value  $-1$  or  $1$  is used it is essential to test the value of IFAIL on exit.**

## 6 Error Indicators and Warnings

If on entry  $IFAIL = 0$  or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

$IFAIL < 0$  and  $IFAIL \neq -999$

If  $IFAIL = -i$ , the  $i$ th argument had an illegal value.

$IFAIL = -999$

Allocation of memory failed. The *complex\*16* allocatable memory required is  $2 \times N$ . In this case the factorization and the solution  $X$  have been computed, but RCOND and ERBND have not been computed.

$IFAIL > 0$  and  $IFAIL \leq N$

If  $IFAIL = i$ ,  $u_{ii}$  is exactly zero. The factorization has been completed, but the factor  $U$  is exactly singular, so the solution could not be computed.

$IFAIL = N + 1$

RCOND is less than *machine precision*, so that the matrix  $A$  is numerically singular. A solution to the equations  $AX = B$  has nevertheless been computed.

## 7 Accuracy

The computed solution for a single right-hand side,  $\hat{x}$ , satisfies an equation of the form

$$(A + E)\hat{x} = b,$$

where

$$\|E\|_1 = O(\epsilon)\|A\|_1$$

and  $\epsilon$  is the *machine precision*. An approximate error bound for the computed solution is given by

$$\frac{\|\hat{x} - x\|_1}{\|x\|_1} \leq \kappa(A) \frac{\|E\|_1}{\|A\|_1},$$

where  $\kappa(A) = \|A^{-1}\|_1 \|A\|_1$ , the condition number of  $A$  with respect to the solution of the linear equations. F04CCF uses the approximation  $\|E\|_1 = \epsilon \|A\|_1$  to estimate ERBND. See Section 4.4 of Anderson *et al.* (1999) for further details.

## 8 Further Comments

The total number of floating-point operations required to solve the equations  $AX = B$  is proportional to  $nr$ . The condition number estimation typically requires between four and five solves and never more than eleven solves, following the factorization.

In practice the condition number estimator is very reliable, but it can underestimate the true condition number; see Section 15.3 of Higham (2002) for further details.

The real analogue of F04CCF is F04BCF.

## 9 Example

To solve the equations

$$AX = B,$$

where  $A$  is the tridiagonal matrix

$$A = \begin{pmatrix} -1.3 + 1.3i & 2.0 - 1.0i & 0 & 0 & 0 \\ 1.0 - 2.0i & -1.3 + 1.3i & 2.0 + 1.0i & 0 & 0 \\ 0 & 1.0 + 1.0i & -1.3 + 3.3i & -1.0 + 1.0i & 0 \\ 0 & 0 & 2.0 - 3.0i & -0.3 + 4.3i & 1.0 - 1.0i \\ 0 & 0 & 0 & 1.0 + 1.0i & -3.3 + 1.3i \end{pmatrix}$$

and

$$B = \begin{pmatrix} 2.4 - 5.0i & 2.7 + 6.9i \\ 3.4 + 18.2i & -6.9 - 5.3i \\ -14.7 + 9.7i & -6.0 - 0.6i \\ 31.9 - 7.7i & -3.9 + 9.3i \\ -1.0 + 1.6i & -3.0 + 12.2i \end{pmatrix}.$$

An estimate of the condition number of  $A$  and an approximate error bound for the computed solutions are also printed.

### 9.1 Program Text

**Note:** the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      F04CCF Example Program Text
*      Mark 21 Release. NAG Copyright 2004.
*      .. Parameters ..
INTEGER          NIN, NOUT
PARAMETER       (NIN=5,NOUT=6)
INTEGER          NMAX, NRHSMX
PARAMETER       (NMAX=8,NRHSMX=2)
INTEGER          LDB
PARAMETER       (LDB=NMAX)
*      .. Local Scalars ..
DOUBLE PRECISION ERBND, RCOND
INTEGER          I, IERR, IFAIL, J, N, NRHS
*      .. Local Arrays ..
COMPLEX *16      B(LDB,NRHSMX), D(NMAX), DL(NMAX-1), DU(NMAX-1),
+               DU2(NMAX-2)
INTEGER          IPIV(NMAX)
CHARACTER        CLABS(1), RLABS(1)
*      .. External Subroutines ..
EXTERNAL         F04CCF, X04DBF
*      .. Executable Statements ..
WRITE (NOUT,*) 'F04CCF Example Program Results'
WRITE (NOUT,*)
*      Skip heading in data file
READ (NIN,*)
READ (NIN,*) N, NRHS
IF (N.LE.NMAX .AND. NRHS.LE.NRHSMX) THEN
*
*      Read A and B from data file
*
READ (NIN,*) (DU(I),I=1,N-1)
READ (NIN,*) (D(I),I=1,N)
READ (NIN,*) (DL(I),I=1,N-1)
READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,N)
*
*      Solve the equations AX = B for X
*
IFAIL = -1
CALL F04CCF(N,NRHS,DL,D,DU,DU2,IPIV,B,LDB,RCOND,ERBND,IFAIL)
```

```

*
      IF (IFAIL.EQ.0) THEN
*
*       Print solution, estimate of condition number and approximate
*       error bound
*
      IERR = 0
      CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed',' ',
+               'Solution','Integer',RLABS,'Integer',CLABS,80,0,
+               IERR)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Estimate of condition number'
      WRITE (NOUT,99999) 1.0D0/RCOND
      WRITE (NOUT,*)
      WRITE (NOUT,*)
+       'Estimate of error bound for computed solutions'
      WRITE (NOUT,99999) ERRBND
      ELSE IF (IFAIL.EQ.N+1) THEN
*
*       Matrix A is numerically singular. Print estimate of
*       reciprocal of condition number and solution
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Estimate of reciprocal of condition number'
      WRITE (NOUT,99999) RCOND
*
      WRITE (NOUT,*)
      IERR = 0
      CALL X04DBF('General',' ',N,NRHS,B,LDB,'Bracketed',' ',
+               'Solution','Integer',RLABS,'Integer',CLABS,80,0,
+               IERR)
*
      ELSE IF (IFAIL.GT.0 .AND. IFAIL.LE.N) THEN
*
*       The upper triangular matrix U is exactly singular. Print
*       details of factorization
*
      WRITE (NOUT,*) 'Details of factorization'
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Second super-diagonal of U'
      WRITE (NOUT,99998) (DU2(I),I=1,N-2)
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' First super-diagonal of U'
      WRITE (NOUT,99998) (DU(I),I=1,N-1)
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Main diagonal of U'
      WRITE (NOUT,99998) (D(I),I=1,N)
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Multipliers'
      WRITE (NOUT,99998) (DL(I),I=1,N-1)
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' Vector of interchanges'
      WRITE (NOUT,99997) (IPIV(I),I=1,N)
      END IF
      ELSE
      WRITE (NOUT,*) 'NMAX and/or NRHSMX too small'
      END IF
      STOP
*
99999 FORMAT (8X,1P,E9.1)
99998 FORMAT (4(1X,'(',F7.4,',',F7.4,')',:))
99997 FORMAT (1X,8I9)
      END

```

## 9.2 Program Data

F04CCF Example Program Data

```

      5              2                                :Values of N and NRHS
(  2.0, -1.0) (  2.0,  1.0) ( -1.0,  1.0) (  1.0, -1.0) :End of DU
( -1.3,  1.3) ( -1.3,  1.3) ( -1.3,  3.3) ( -0.3,  4.3)
( -3.3,  1.3)                                         :End of D
(  1.0, -2.0) (  1.0 , 1.0) (  2.0, -3.0) (  1.0,  1.0) :End of DL
(  2.4, -5.0) (  2.7,  6.9)
(  3.4, 18.2) ( -6.9, -5.3)
(-14.7,  9.7) ( -6.0, -0.6)
( 31.9, -7.7) ( -3.9,  9.3)
( -1.0,  1.6) ( -3.0, 12.2)                             :End of B

```

## 9.3 Program Results

F04CCF Example Program Results

Solution

```

                                1                2
1 (  1.0000,  1.0000) (  2.0000, -1.0000)
2 (  3.0000, -1.0000) (  1.0000,  2.0000)
3 (  4.0000,  5.0000) ( -1.0000,  1.0000)
4 ( -1.0000, -2.0000) (  2.0000,  1.0000)
5 (  1.0000, -1.0000) (  2.0000, -2.0000)

```

Estimate of condition number  
1.8E+02

Estimate of error bound for computed solutions  
2.0E-14

---